One-loop calculations with FeynCalc

Vladyslav Shtabovenko

in collaboration with R. Mertig and F. Orellana

Technische Universität München

IMPRS Young Scientist Workshop at
Ringberg Castle 2015

Technische Universität München



Physik-Department T30f

Outline

1. Motivation
   - QFT Automation
   - FeynCalc
2. Using FeynCalc in your research
   - 1-loop calculations
   - Example: Schwinger's triumph
3. Summary and outlook

What

- Automation of QFT $\approx$ automatic symbolic or numeric evaluation of Feynman diagrams
- This talks: only symbolics

Why

- Feynman diagrams $\rightarrow$ theoretical predictions for experimental observables: cross-sections, decay rates etc.
- Experiments are becoming more precise $\rightarrow$ theorists must reduce errors in their predictions to keep up.
- Leading order (LO) in perturbation theory is mostly not enough, need to go at least to NLO or even higher
- Need to evaluate hundreds, thousands or even millions of Feynman diagrams
- Impossible to do by pen and paper!

How

- CAS or CAS-like environment: Mathematica, Reduce, FORM, Sympy, GiNaC etc.
- Specific codes running on top of it.

Automation tools classified by their usage

- Single purpose tools: FeynArts[1], Tracer [2], FIRE[3], LoopTools[4], ...
- Multi purpose tools (semi-automatic): HEPMath[5], FeynCalc, Package X[6], ...
- Multi purpose tools (fully-automatic): CalcHEP[7], GRACE[8], FormCalc[1] ...



[xkcd.com/1319/]

[1] [Hahn, 2001]
[2] [Jamin & Lautenbacher, 1993]
[3] [Smirnov, 2008]
[4] [Hahn & Perez-Victoria, 1999]
[5] [Wiebusch, 2014]
[6] [Patel, 2015]
[7] [Belyaev et al., 2012]
[8] [Ishikawa et al., 1993]

- FeynCalc is a Mathematica package for algebraic QFT calculations

  [Mertig et al., 1991]

- Suitable for evaluating both single expression and full Feynman diagrams

Feyn
Calc

Features

- Extensive typesetting for better readability

  (using Mathematica's `TraditionalForm` output)

- Tools for frequently occurring tasks like Lorentz index contraction, SU(N) algebra, Dirac matrix manipulation and traces, etc.

  (`Contract`, `SUNSimplify`, `SUNTrace`, `DiracSimplify`, `DiracTrace`, `DiracEquation`, `DiracReduce`, `Schouten`)

- Passarino-Veltman reduction of one-loop amplitudes to standard scalar integrals

  (`OneLoop`, `OneLoopSimplify`, `TID`, `Tdec`, `PaVeReduce`, `ScalarProductCancel`, `FeynAmpDenominatorSimplify`, `FCLoop*`)

- General tools for non-commutative algebra

  (`DotSimplify`, `DotExpand`, `DeclareNonCommutative`, `UnDeclareNonCommutative`, `Commutator`, `Anticommutator`)

- The calculation can be organized in many different ways (flexibility)

Challenges

- Very little development between 2006 and 2014
- Bugs
- The performance of many functions is not optimal

Improvements since 2014

- New collaborator (VS)
- Numerous bugfixes
- Prevent regressions by introducing unit and integration test ($\approx$ 3000 tests so far)
- To improve performance and stability some functions (`DiracTrick`, `DiracEquation`, `Anti5`, `TID`, `Tdec`, ...) were rewritten almost from scratch
- Code is now hosted on GitHub: `github.com/FeynCalc`
- Lots of new examples (mostly QED and QCD) included
- FeynCalc wiki: `https://github.com/FeynCalc/feyncalc/wiki`

Ingredients of a 1-loop calculation in dimensional regularization (DR)

$$\int \frac{d^4\bar{l}}{(2\pi)^4} \frac{(\bar{l}^\mu \bar{l}^\nu)}{\bar{l}^2 - m^2} \to \mu^{D-4} \int \frac{d^D l}{(2\pi)^D} \frac{(l^\mu l^\nu)}{l^2 - m^2}$$

- Simplification of the Dirac algebra (✓FeynCalc)
- Reduction of tensor integrals to scalar integrals
  - Cancellation of scalar products and (✓FeynCalc)
  - Tensor decomposition (✓FeynCalc)
- Further simplification of scalar integrals
  - Partial fractioning (✓FeynCalc)
  - IBP reduction (usually requires external tools)
- Evaluation of master integrals (requires external tools)

Why Dirac algebra?

In Feynman diagrams with internal fermion lines, loop momenta are often contracted with the Dirac matrices. For example,

$$\int \frac{d^D l}{(2\pi)^D} \frac{\gamma_\nu \gamma_\mu l^\nu l^\mu}{l^2((l+p)^2 - m^2)}$$

Naive solution: Ignore the Dirac matrices. Just uncontract loop momenta and simplify the resulting tensor integrals

In[1]:= GSD[l].GSD[l] FAD[l,{l+p,m}]

Out[1]:= $(\gamma \cdot l).(\gamma \cdot l) \frac{1}{([l^2])\,([(l+p)^2 - m^2])}$

In[2]:= Uncontract[%%,l]//FCLoopIsolate[#,{l}]&

Out[2]:= $\gamma^{\$AL\$2494(1)}.\gamma^{\$AL\$2494(2)}\mathrm{FCGV}(\mathrm{LoopInt})\left(\frac{l^{\$AL\$2494(1)}\, l^{\$AL\$2494(2)}}{l^2.\,((l+p)^2 - m^2)}\right)$

Clever solution: First simplify the Dirac algebra, then reduce the integrals

In[3]:= GSD[l].GSD[l] FAD[l,{l+p,m}];
In[4]:= DiracSimplify[%,l]

Out[4]:= $\frac{l^2}{l^2.\,((l+p)^2 - m^2)}$

Dirac algebra

The generalization of the Dirac algebra to D dimensions is (almost) straight-forward ['t Hooft & Veltman, 1972]

$$\{\bar{\gamma}^\mu, \bar{\gamma}^\nu\} = 2\bar{g}^{\mu\nu}, \quad \bar{g}^{\mu\nu}\bar{g}_{\mu\nu} = 4 \rightarrow \{\gamma^\mu, \gamma^\nu\} = 2g^{\mu\nu}, \quad g^{\mu\nu}g_{\mu\nu} = D$$

So we can easily simplify the algebra and compute traces in D-dimensions

---

In[1]:= GAD[μ,ν]
Out[1]= $\gamma^\mu.\gamma^\mu$
In[2]:= %//DiracSimplify
Out[2]= D

In[3]:= GSD[l,l]
Out[3]= $(\gamma \cdot l).(\gamma \cdot l)$
In[4]:= %//DiracSimplify
Out[4]= l^2

In[5]:= GSD[l].GAD[μ].GSD[l]
Out[5]= $(\gamma \cdot l).\gamma^\mu.(\gamma \cdot l)$
In[6]:= %//DiracSimplify
Out[6]= $2l^\mu\gamma \cdot l - l^2\gamma^\mu$

In[7]:= DiracTrace[GAD[μ].GSD[p].GAD[ν].GSD[p]]
Out[7]= tr $\left(\gamma^\mu.(\gamma \cdot p).\gamma^\nu.(\gamma \cdot p)\right)$
In[8]:= % /. DiracTrace $->$ **Tr**
Out[8]= $4\left(2p^\mu p^\nu - p^2 g^{\mu\nu}\right)$

Cancelling scalar products

Before starting to reduce a tensor integral, it is always useful to first try to cancel as much scalar products as possible using the well-known trick

$$l \cdot k = \frac{1}{2}(l+k)^2 - \frac{1}{2}l^2 - \frac{1}{2}k^2$$

In[1]:= SPD[l, p1] FVD[l, μ] FAD[{l, m0}, {l + p1, m1}, {l + p2, m2}]

Out[1]= $l^{\mu}(l \cdot \mathrm{p1}) \frac{1}{([l^2 - \mathrm{m0}^2])([(l+\mathrm{p1})^2 - \mathrm{m1}^2])([(l+\mathrm{p2})^2 - \mathrm{m2}^2])}$

In[2]:= SPC[#,l]

Out[2]=

$$\frac{l^{\mu}\left(\mathrm{m0}^2 - \mathrm{m1}^2\right)}{2\,(l^2 - \mathrm{m0}^2)\,.\,((l-\mathrm{p1})^2 - \mathrm{m1}^2)\,.\,((l-\mathrm{p2})^2 - \mathrm{m2}^2)} - \frac{l^{\mu}}{2\,(l^2 - \mathrm{m0}^2)\,.\,((l-\mathrm{p2})^2 - \mathrm{m2}^2)}$$

$$-\frac{l^{\mu}}{2\,(l^2 - \mathrm{m1}^2)\,.\,((l-\mathrm{p1}+\mathrm{p2})^2 - \mathrm{m2}^2)} + \frac{\mathrm{p1}^{\mu}}{2\,(l^2 - \mathrm{m1}^2)\,.\,((l-\mathrm{p1}+\mathrm{p2})^2 - \mathrm{m2}^2)}$$

Passarino Veltman reduction

- Passarino-Veltman reduction is the standard technique for the tensor decomposition of loop integrals.

  [t'Hooft & Veltman, 1979]
  [Passarino & Veltman, 1979]
- Lorentz covariance allows us to rewrite any tensor integral as a linear combination of all allowed Lorentz structures
- These structures are made of metric tensors and external momenta
- They are also multiplied by scalar coefficients. These coefficients (aka Passarino-Veltman coefficient functions) can be computed either analytically or numerically.

$$\int \frac{d^D l}{(2\pi)^D} \frac{l^\mu l^\nu}{[l^2 - m^2][(l+p)^2 - m^2]} = g^{\mu\nu} B_{00} + p^\mu p^\nu B_{11}$$

Contracting with $g^{\mu\nu}$ and $p^\mu p^\nu$ we obtain a linear system of scalar equations

$$\int \frac{d^D l}{(2\pi)^D} \frac{l^2}{[l^2 - m^2][(l+p)^2 - m^2]} = D B_{00} + p^2 B_{11}$$
$$\int \frac{d^D l}{(2\pi)^D} \frac{(l \cdot p)^2}{[l^2 - m^2][(l+p)^2 - m^2]} = p^2 B_{00} + p^4 B_{11}$$

Solving this system we can determine the coefficients B_{00} and B_{11}.

We can of course do this also with FeynCalc

In[1]:= FVD[l, μ] FVD[l, ν] FAD[{l, m0}]

Out[1]= $l^{\mu} l^{\nu} \frac{1}{[l^2 - \mathrm{m0}^2]}$

In[2]:= TID[1/(**I Pi**^2) %, l, UsePaVeBasis −> **True**]

Out[2]= $\frac{\mathrm{m0}^2 \mathrm{A}_0 \left(\mathrm{m0}^2\right) g^{\mu\nu}}{D}$

In[3]:= FVD[l, μ] FAD[{l, m0}, {l + p1, m1}]

Out[3]:= $l^{\mu} \frac{1}{([l^2 - \mathrm{m0}^2]) ([(l + \mathrm{p1})^2 - \mathrm{m1}^2])}$

In[4]:= TID[1/(**I Pi**^2) %, l, UsePaVeBasis −> **True**, PaVeAutoReduce −> **False**]

Out[4]:= $\mathrm{p1}^{\mu} \mathrm{B}_1 \left(\mathrm{p1}^2, \mathrm{m0}^2, \mathrm{m1}^2\right)$

also for more complicated integrals

In[1]:= FVD[l, μ] FVD[l, ν] FVD[
l, ρ] FAD[{l, m0}, {l + p1, m1}, {l + p2, m2}]

Out[1]= $l^{\mu} l^{\nu} l^{\rho} \frac{1}{([l^2 - \text{m0}^2])([(l+\text{p1})^2 - \text{m1}^2])([(l+\text{p2})^2 - \text{m2}^2])}$

In[2]:= TID[1/(**I Pi**^2) %, l, UsePaVeBasis −> **True**, PaVeAutoReduce −> **False**]

Out[2]=

$$\left(\text{p1}^{\mu} g^{\nu\rho} + \text{p1}^{\nu} g^{\mu\rho} + \text{p1}^{\rho} g^{\mu\nu}\right) \text{C}_{001}\left(\text{p1}^2, -2(\text{p1}\cdot\text{p2}) + \text{p1}^2 + \text{p2}^2, \text{p2}^2, \text{m0}^2, \text{m1}^2, \text{m2}^2\right)$$
$$+ \left(\text{p2}^{\mu} g^{\nu\rho} + \text{p2}^{\nu} g^{\mu\rho} + \text{p2}^{\rho} g^{\mu\nu}\right) \text{C}_{002}\left(\text{p1}^2, -2(\text{p1}\cdot\text{p2}) + \text{p1}^2 + \text{p2}^2, \text{p2}^2, \text{m0}^2, \text{m1}^2, \text{m2}^2\right)$$
$$+\text{p1}^{\mu}\text{p1}^{\nu}\text{p1}^{\rho}\text{C}_{111}\left(\text{p1}^2, -2(\text{p1}\cdot\text{p2}) + \text{p1}^2 + \text{p2}^2, \text{p2}^2, \text{m0}^2, \text{m1}^2, \text{m2}^2\right)$$
$$+ \left(\text{p1}^{\nu}\text{p1}^{\rho}\text{p2}^{\mu} + \text{p1}^{\mu}\text{p1}^{\rho}\text{p2}^{\nu} + \text{p1}^{\mu}\text{p1}^{\nu}\text{p2}^{\rho}\right)$$
$$\times\text{C}_{112}\left(\text{p1}^2, -2(\text{p1}\cdot\text{p2}) + \text{p1}^2 + \text{p2}^2, \text{p2}^2, \text{m0}^2, \text{m1}^2, \text{m2}^2\right)$$
$$+ \left(\text{p1}^{\rho}\text{p2}^{\mu}\text{p2}^{\nu} + \text{p1}^{\nu}\text{p2}^{\mu}\text{p2}^{\rho} + \text{p1}^{\mu}\text{p2}^{\nu}\text{p2}^{\rho}\right)$$
$$\times\text{C}_{122}\left(\text{p1}^2, -2(\text{p1}\cdot\text{p2}) + \text{p1}^2 + \text{p2}^2, \text{p2}^2, \text{m0}^2, \text{m1}^2, \text{m2}^2\right)$$
$$+\text{p2}^{\mu}\text{p2}^{\nu}\text{p2}^{\rho}\text{C}_{222}\left(\text{p1}^2, -2(\text{p1}\cdot\text{p2}) + \text{p1}^2 + \text{p2}^2, \text{p2}^2, \text{m0}^2, \text{m1}^2, \text{m2}^2\right)$$

Partial fractioning

Even after all tensor integrals have been decomposed to scalar ones, partial fractioning allows us to simplify some of the even further

In[1]:= FAD[{p,m1},{p,m2}]

Out[1]= $\frac{1}{([p^2 - \text{m1}^2])\,([p^2 - \text{m2}^2])}$

In[2]:= Apart2[%]//**Expand**

Out[2]= $\frac{1}{(\text{m1}^2 - \text{m2}^2)\,(p^2 - \text{m1}^2)} - \frac{1}{(\text{m1}^2 - \text{m2}^2)\,(p^2 - \text{m2}^2)}$

As long as the kinematics inside the loop integral is general, we can write it in terms of the 4 Passarino Veltman basis integrals $A0$, $B0$, $C0$ and $D0$.

In[1]:= ClearScalarProducts;
ScalarProduct[p1, p2] = 2 M^2;
ScalarProduct[p1, p1] = M^2;
ScalarProduct[p2, p2] = M^2;
FVD[l, \[Mu]] FAD[{l, m0}, {l + p1, m1}, {l + p2, m2}];

Out[1]:= $$l^\mu \frac{1}{([l^2 - \mathrm{m0}^2])\,([(l+\mathrm{p1})^2 - \mathrm{m1}^2])\,([(l+\mathrm{p2})^2 - \mathrm{m2}^2])}$$

In[2]:= TID[%, l] // Isolate [#, FeynAmpDenominator] & //
ReplaceAll[#, FeynAmpDenominator[x___] :>FRH[FeynAmpDenominator[x]]]&

Out[2]:= $$\frac{\mathrm{KK}(147)}{6M^4\,(l^2-\mathrm{m0}^2)\,.\,((l-\mathrm{p1})^2-\mathrm{m1}^2)\,.\,((l-\mathrm{p2})^2-\mathrm{m2}^2)}$$
$$+\frac{\mathrm{KK}(141)}{6M^4\,(l^2-\mathrm{m1}^2)\,.\,((l-\mathrm{p1})^2-\mathrm{m0}^2)}+\frac{\mathrm{KK}(143)}{6M^4\,(l^2-\mathrm{m0}^2)\,.\,((l-\mathrm{p1})^2-\mathrm{m1}^2)}$$
$$-\frac{\mathrm{KK}(141)}{6M^4\,(l^2-\mathrm{m2}^2)\,.\,((l-\mathrm{p2})^2-\mathrm{m0}^2)}-\frac{\mathrm{KK}(143)}{6M^4\,(l^2-\mathrm{m2}^2)\,.\,((l-\mathrm{p1}+\mathrm{p2})^2-\mathrm{m1}^2)}$$

But a special choice of the kinematics might lead to vanishing Gram determinants. Such tensor integrals are then written in terms of the coefficient functions.

In[1]:= ClearScalarProducts;
ScalaProduct[p1, p2] = 0;
ScalarProduct[p1, p1] = 0;
ScalarProduct[p2, p2] = 0;
FVD[l, \[Mu]] FAD[{l, m0}, {l + p1, m1}, {l + p2, m2}];

Out[1]:= $l^{\mu} \frac{1}{([l^2 - \mathrm{m0}^2])([(l+\mathrm{p1})^2 - \mathrm{m1}^2])([(l+\mathrm{p2})^2 - \mathrm{m2}^2])}$

In[2]:= TID[1/(**I Pi**^2) %, l]

Out[2]:= $-\mathrm{p2}^{\mu}\mathrm{C}_1\left(0,0,0,\mathrm{m2}^2,\mathrm{m0}^2,\mathrm{m1}^2\right) + \left(\mathrm{p1}^{\mu} - \mathrm{p2}^{\mu}\right)\mathrm{C}_2\left(0,0,0,\mathrm{m2}^2,\mathrm{m0}^2,\mathrm{m1}^2\right)$
$+ \frac{i\mathrm{p2}^{\mu}}{\pi^2\left(l^2-\mathrm{m0}^2\right).\left((l-\mathrm{p1})^2-\mathrm{m1}^2\right).\left((l-\mathrm{p2})^2-\mathrm{m2}^2\right)}$

In[3]:= % // ToPaVe[#, l] &

Out[3]:= $\mathrm{p2}^{\mu}\left(-\mathrm{C}_0\left(0,0,0,\mathrm{m0}^2,\mathrm{m1}^2,\mathrm{m2}^2\right)\right) - \mathrm{p2}^{\mu}\mathrm{C}_1\left(0,0,0,\mathrm{m2}^2,\mathrm{m0}^2,\mathrm{m1}^2\right)$
$+ \left(\mathrm{p1}^{\mu} - \mathrm{p2}^{\mu}\right)\mathrm{C}_2\left(0,0,0,\mathrm{m2}^2,\mathrm{m0}^2,\mathrm{m1}^2\right)$

Feyncalc can algebraically simplify many standalone QFT expressions. What about Feynman diagrams?

Generating Feynman diagrams

- FeynCalc itself can't generate any diagrams

  $\Rightarrow$ Use FeynArts
- Some objects in FeynCalc and FeynArts have same names (e.g. `FourVector`) which leads to issues

  $\Rightarrow$ Patch FeynArts to rename conflicting objects
- The output of FeynArts is incompatible with FeynArts

  Convert it to FeynCalc via `FCPrepareFAAmp`

To see how FeynArts+FeynCalc can be used to evaluate Feynman diagrams, let us calculate the anomalous electric moment of the electron at 1-loop in QED.

g-2: Short reminder

- g is the coupling of the electron to the magnetic field in the non-relativistic limit

$$V(x) = -\vec{\mu} \cdot \vec{B}(x), \quad \vec{\mu} = \frac{g}{2m_e} \vec{S}$$

- Expanding the Dirac equation

$$(i\gamma_\mu D^\mu - m_e)\psi = 0$$

  in $1/m_e$ yields $g = 2$.
- However, this is just a tree-level result. Loop corrections induce an anomalous electric moment with $g - 2 \neq 0$
- To extract the value of electron's $g - 2$ to $\mathcal{O}(\alpha)$ in QED we need to consider the 1-loop correction to the electron-photon vertex.



We have

$$i\mathcal{M}^\mu = -ie\bar{u}(p_2)\left(\gamma^\mu(F_1(k^2)+F_2(k^2)) - \frac{1}{2m_e}(p_1+p_2)^\mu F_2(k^2)\right)u(p_1),$$

with

$$g = 2 + 2F_2(0)$$

So our goal is to extract the F_2 from $i\mathcal{M}^\mu$ at zero momentum transfer.

We start with loading FeynCalc and FeynArts

```
In[1]:= $LoadFeynArts=True;
        <<FeynCalc`
        $FAVerbose=0;
```

FeynCalc 9.0.0. For help, type ?FeynCalc, use the `helpbrowser` or visit www.feyncalc.org.
FeynArts 3.9 patched for use with FeynCalc, for documentation use the manual or visit www.feynarts.de.

Then we use FeynArts to generate our diagram

```
In[2]:= topVertex = CreateTopologies[1, 1 -> 2, ExcludeTopologies -> {Tadpoles, WFCorrections
      }];
In[3]:= diagsVertex = InsertFields [topVertex, {F[2, {1}]} -> {V[1], F[2, {1}]}, InsertionLevel
      -> {Classes}, Model -> "SM", ExcludeParticles -> {S[1], S[2], S[3], V[3], V[2]}];
In[4]:= Paint[diagsVertex, ColumnsXRows -> {2, 1}, Numbering -> None]
```



Next step is to convert the output of FeynArts into input for FeynCalc

In[5]:= ampVertex = Total@Map[**ReplaceAll**[#, FeynAmp[_, _, amp_, ___] :> amp] &,
Apply[**List** , FCPrepareFAAmp[CreateFeynAmp[diagsVertex, Truncated −> **False**,
PreFactor −> −1], UndoChiralSplittings −> **True**]]] /.
{InMom1 −> p1, OutMom2 −> p2, OutMom1 −> k, LoopMom1 −> q} /.
k −> p1 − p2 /. q −> q + p1;
Out[5]:=

$$-\left(i\bar{g}^{\mathrm{Lor2Lor3}}\bar{\varepsilon}^{*\mathrm{Lor1}}(\mathrm{p1}-\mathrm{p2})\left(\varphi(\overline{\mathrm{p2}},\mathrm{ME})\right).\left(i\mathrm{EL}\bar{\gamma}^{\mathrm{Lor3}}\right).\left(\bar{\gamma}\cdot\left(\overline{\mathrm{p2}}+\overline{q}\right)+\mathrm{ME}\right).\left(i\mathrm{EL}\bar{\gamma}^{\mathrm{Lor1}}\right).\left(\bar{\gamma}\cdot\left(\overline{\mathrm{p1}}+\overline{q}\right)+\mathrm{ME}\right).\left(i\mathrm{EL}\bar{\gamma}^{\mathrm{Lor2}}\right).\left(\varphi(\overline{\mathrm{p1}},\mathrm{ME})\right)\right)/\left((\mathrm{p1}+q)^2-\mathrm{ME}^2\right).\left((\mathrm{p2}+q)^2-\mathrm{ME}^2\right).q^2$$

after which we set up the kinematics, convert the obtained amplitude into a D-dimensional one and chop off the polarization vector

In[6]:= ClearScalarProducts ;
In[7]:= ScalarProduct[p1, p1] = ME^2;
In[8]:= ScalarProduct[p2, p2] = ME^2;
In[9]:= ScalarProduct[k, k] = 0;
In[10]:= ScalarProduct[p1, p2] = ME^2;
In[11]:= ampVertex1 = (ampVertex // ChangeDimension[#, **D**] &)//
ReplaceAll[#, Pair[Momentum[Polarization[___], ___], ___] :> 1] &;

Now we simplify whatever we can simplify and reduce our tensor loop integrals into scalar ones.

In[12]:= OneLoopSimplify[ampVertex1, q] // Collect2[#, Spinor] &

$$\mathbf{Out}[12]{:=}\ 2i\pi^2\mathsf{EL}^3\mathsf{ME}\left(\mathsf{p1}^{\mathsf{Lor1}}+\mathsf{p2}^{\mathsf{Lor1}}\right)\Big(2\mathsf{C}_1\left(\mathsf{ME}^2,0,\mathsf{ME}^2,0,\mathsf{ME}^2,\mathsf{ME}^2\right)+ D\mathsf{C}_{11}\left(\mathsf{ME}^2,0,\mathsf{ME}^2,0,\mathsf{ME}^2,\mathsf{ME}^2\right)-2\mathsf{C}_{11}\left(\mathsf{ME}^2,0,\mathsf{ME}^2,0,\mathsf{ME}^2,\mathsf{ME}^2\right)+ +D\mathsf{C}_{12}\left(\mathsf{ME}^2,0,\mathsf{ME}^2,0,\mathsf{ME}^2,\mathsf{ME}^2\right)-2\mathsf{C}_{12}\left(\mathsf{ME}^2,0,\mathsf{ME}^2,0,\mathsf{ME}^2,\mathsf{ME}^2\right)\Big) (\varphi(\mathsf{p2},\mathsf{ME})).(\varphi(\mathsf{p1},\mathsf{ME}))-\mathsf{EL}^3\Big(-2i\pi^2 D\mathsf{C}_{00}\left(\mathsf{ME}^2,0,\mathsf{ME}^2,0,\mathsf{ME}^2,\mathsf{ME}^2\right) +4i\pi^2\mathsf{C}_{00}\left(\mathsf{ME}^2,0,\mathsf{ME}^2,0,\mathsf{ME}^2,\mathsf{ME}^2\right)+\frac{D}{(q^2-\mathsf{ME}^2).((-\mathsf{p1}+\mathsf{p2}+q)^2-\mathsf{ME}^2)} +\frac{4\mathsf{ME}^2}{q^2.((q-\mathsf{p1})^2-\mathsf{ME}^2).((q-\mathsf{p2})^2-\mathsf{ME}^2)}-\frac{6}{(q^2-\mathsf{ME}^2).((-\mathsf{p1}+\mathsf{p2}+q)^2-\mathsf{ME}^2)} +\frac{2}{q^2.((q-\mathsf{p1})^2-\mathsf{ME}^2)}+\frac{2}{q^2.((q-\mathsf{p2})^2-\mathsf{ME}^2)}\Big)(\varphi(\mathsf{p2},\mathsf{ME})).\gamma^{\mathsf{Lor1}}.(\varphi(\mathsf{p1},\mathsf{ME}))$$

Remember that to extract $F_2(0)$ we need to look only at the piece proportional to $(p_1+p_2)^\mu$. So let us drop the γ^μ-piece

In[13]:= ampVertex3 = ampVertex2 // **ReplaceAll**[#,FCI[GAD[Lor1]] :> 0] & // DotSimplify

Out[13]:= $2i\pi^2 \mathrm{EL}^3 \mathrm{ME} \left(\mathrm{p1}^{\mathrm{Lor1}} + \mathrm{p2}^{\mathrm{Lor1}}\right) \left(2\mathrm{C}_1\left(\mathrm{ME}^2, 0, \mathrm{ME}^2, 0, \mathrm{ME}^2, \mathrm{ME}^2\right) + D\mathrm{C}_{11}\left(\mathrm{ME}^2, 0, \mathrm{ME}^2, 0, \mathrm{ME}^2, \mathrm{ME}^2\right) - 2\mathrm{C}_{11}\left(\mathrm{ME}^2, 0, \mathrm{ME}^2, 0, \mathrm{ME}^2, \mathrm{ME}^2\right) + {} + D\mathrm{C}_{12}\left(\mathrm{ME}^2, 0, \mathrm{ME}^2, 0, \mathrm{ME}^2, \mathrm{ME}^2\right) - 2\mathrm{C}_{12}\left(\mathrm{ME}^2, 0, \mathrm{ME}^2, 0, \mathrm{ME}^2, \mathrm{ME}^2\right)\right) (\varphi(\mathrm{p2}, \mathrm{ME})).(\varphi(\mathrm{p1}, \mathrm{ME}))$

The Passarino-Veltman coefficient functions C_1, C_{11} and C_{12} that appear in the result can be analytically evaluated using other packages (e.g. Package X). Here we just substitute their values

In[14]:= ampVertex4 = ampVertex3 /. {
PaVe[1, {ME^2, 0, ME^2}, {0, ME^2, ME^2}, OptionsPattern[]] −> 1/(32 **Pi**^4 ME^2),
PaVe[1, 1, {ME^2, 0, ME^2}, {0, ME^2, ME^2}, OptionsPattern[]] −> −(1/(96 **Pi**^4 ME^2)),
PaVe[1, 2, {ME^2, 0, ME^2}, {0, ME^2, ME^2}, OptionsPattern[]] −> −(1/(192 **Pi**^4 ME^2))}

Out[14]:= $2i\pi^2 \mathrm{EL}^3 \mathrm{ME} \left(\frac{3}{32\pi^4 \mathrm{ME}^2} - \frac{D}{64\pi^4 \mathrm{ME}^2}\right) \left(\mathrm{p1}^{\mathrm{Lor1}} + \mathrm{p2}^{\mathrm{Lor1}}\right) (\varphi(\mathrm{p2}, \mathrm{ME})).(\varphi(\mathrm{p1}, \mathrm{ME}))$

As expected, $F_2(0)$ is free of any divergences. So we can safely do the limit $D \to 4$

In[15]:= ampVertex5 = ampVertex4 // ChangeDimension[#, 4] & // **ReplaceAll**[#, **D** −> 4] &

Out[15]:= $\frac{i\mathrm{EL}^3 \left(\overline{\mathrm{p1}}^{\mathrm{Lor1}} + \overline{\mathrm{p2}}^{\mathrm{Lor1}}\right) \left(\varphi(\mathrm{ME}, \overline{\mathrm{p2}})\right) . \left(\varphi(\mathrm{ME}, \overline{\mathrm{p1}})\right)}{16\pi^2 \mathrm{ME}}$

What we obtained so far is nothing else than $\frac{ie}{2m_e}(p_1 + p_2)^\mu F_2(0)\bar{u}(p_2)u(p_1)$.
Dividing by the numerical prefactor and substituting $e^2 = 4\pi^2\alpha$ yields

In[16]:= (ampVertex5/((**I** EL)/(2 ME))) // **ReplaceAll** [#, {
EL^2 −> AlphaFS 4 \[**Pi**], Spinor[___].Spinor[___] :> 1, FCI[FV[p1, _] + FV[p2, _]] :> 1}] &

Out[16]:= $\frac{\alpha}{2\pi}$

In[17]:= **N**[1/137 1/(2 **Pi**)]
Out[17]:= 0.00116171

so that

$$F_2(0) = \frac{\alpha}{2\pi}$$

and

$$\frac{g-2}{2} = \frac{\alpha}{2\pi} + \mathcal{O}(\alpha^2)$$

which was one of the greatest triumphs of QED in the last century

[Schwinger, 1948]

What I learned while developing FeynCalc

General recommendations

- Use a version control system (e.g. git, mercurial)
- Use a testing framework to prevent regressions (e.g. MUnit for Mathematica, pyunit for Python, CppUnit for C++)

Mathematica specific recommendations

- Have a look at Wolfram Workbench
- Read at least one book about Mathematica programming

Summary

- FeynCalc is a Mathematica package for algebraic calculations in QFT and semi-automatic evaluation of Feynman diagrams
- After a long period of low-activity the active development has been restarted 2014
- The upcoming FeynCalc 9.0 will include numerous bug fixes but also performance enhancements and new features

TODOs:

- More regression and integration tests (goal: full code coverage)
- More worked out examples
- Finish the manual
- Stable interfaces to other useful software tools for symbolic/numeric evaluation

Issues with γ^5

- There is no unique way to handle $\gamma^5 = i\gamma^0\gamma^1\gamma^2\gamma^3$ in DR
- In D-dimensions, the relations

$$\{\gamma^5, \gamma^\mu\} = 0$$

and

$$\mathrm{tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma) \neq 0$$

cannot be simultaneously satisfied.

- In other words, there is a conflict between the anticommutativity of γ^5 and the cyclicity property of Dirac traces that involve and odd number of γ^5

[Chanowitz et al., 1979]
[Jegerlehner, 2001]

Issues with γ^5

- We can stick to the anticommuting γ^5 in D-dimensions. This is fine, as long as we have only traces with an even number of γ^5.

  $\Rightarrow$ Naive dimensional regularization (NDR)
- To compute traces with an odd number of γ^5 unambiguously, we need an additional prescription

  $\Rightarrow$ Kreimer's prescription

  [Kreimer, 1990]

  $\Rightarrow$ Larin-Gorishny-Akyeampong-Delburgo prescription

  [Larin, 1993]
- Or we can accept that γ^5 is a purely 4-dimensional object and therefore doesn't anitcommute with D-dimensional Dirac matrices

  ['t Hooft & Veltman, 1972]
  [Breitenlohner & Maison, 1977]

  $\Rightarrow$ Breitenlohner-Maison- t'Hooft Veltman scheme (BMHV)

By default, FeynCalc works with an anticommuting γ^5

In[1]:= GAD[μ,ν,ρ].GA[5].GAD[σ,τ,κ].GA[5]
Out[1]= $\gamma^\mu.\gamma^\nu.\gamma^\rho.\bar{\gamma}^5.\gamma^\sigma.\gamma^\tau.\gamma^\kappa.\bar{\gamma}^5$
In[2]:= %//DiracSimplify
Out[2]= $-\gamma^\mu.\gamma^\nu.\gamma^\rho.\gamma^\sigma.\gamma^\tau.\gamma^\kappa$

Trying to compute a chiral trace in the naive scheme produces an error message:

In[1]:= DiracTrace[GAD[μ,ν,ρ,σ,τ,κ].GA[5]]
Out[1]= tr $\left(\gamma^\mu.\gamma^\nu.\gamma^\rho.\gamma^\sigma.\gamma^\tau.\gamma^\kappa.\bar{\gamma}^5\right)$
In[2]:= % /. DiracTrace −> **Tr**

```
DiracTrace::ndranomaly :
  You are using naive dimensional regularization (NDR), such
    that in D dimensions gamma^5 anticommutes with all
    other Dirac matrices. In this scheme (without additional
    prescriptions) it is not possible to compute traces with
    an odd number of gamma^5 unambiguously. The trace

       DiracGamma[LorentzIndex[μ, D], D].
         DiracGamma[LorentzIndex[ν, D], D].
    of   DiracGamma[LorentzIndex[ρ, D], D].
         DiracGamma[LorentzIndex[σ, D], D].
         DiracGamma[LorentzIndex[τ, D], D].
         DiracGamma[LorentzIndex[κ, D], D].DiracGamma[5]

    is illegal in NDR. Evaluation aborted!
```

D-dimensional traces with anticommuting γ^5 can be evaluated using Larin-Gorishny-Akyeampong-Delburgo prescription

In[1]:= \$Larin = **True**;
In[2]:= \$West = **False**;
In[3]:= \$BreitMaison = **False**;
In[4]:= DiracTrace[GAD[μ, ν, ρ, σ, τ, κ]. GA[5]]
Out[4]:= $\mathrm{tr}\left(\gamma^{\mu}.\gamma^{\nu}.\gamma^{\rho}.\gamma^{\sigma}.\gamma^{\tau}.\gamma^{\kappa}.\bar{\gamma}^{5}\right)$
In[5]:= % /. DiracTrace −> **Tr**
Out[5]:= $4\left(ig^{\mu\nu}\epsilon^{\kappa\rho\sigma\tau} - ig^{\mu\rho}\epsilon^{\kappa\nu\sigma\tau} + ig^{\mu\sigma}\epsilon^{\kappa\nu\rho\tau} - ig^{\mu\tau}\epsilon^{\kappa\nu\rho\sigma} + ig^{\nu\rho}\epsilon^{\kappa\mu\sigma\tau} - ig^{\nu\sigma}\epsilon^{\kappa\mu\rho\tau} + ig^{\nu\tau}\epsilon^{\kappa\mu\rho\sigma} + ig^{\rho\sigma}\epsilon^{\kappa\mu\nu\tau} - ig^{\rho\tau}\epsilon^{\kappa\mu\nu\sigma} + ig^{\sigma\tau}\epsilon^{\kappa\mu\nu\rho}\right)$

Or in the BMHV scheme

In[6]:= \$Larin = **False**;
In[7]:= \$West = **True**;
In[8]:= \$BreitMaison = **False**;
In[9]:= DiracTrace[GAD[μ, ν, ρ, σ, τ, κ]. GA[5]]
Out[9]:= $\mathrm{tr}\left(\gamma^{\mu}.\gamma^{\nu}.\gamma^{\rho}.\gamma^{\sigma}.\gamma^{\tau}.\gamma^{\kappa}.\bar{\gamma}^{5}\right)$
In[10]:= % /. DiracTrace −> **Tr**
Out[10]:= $4\left(-ig^{\kappa\mu}\epsilon^{\nu\rho\sigma\tau} + ig^{\kappa\nu}\epsilon^{\mu\rho\sigma\tau} - ig^{\kappa\rho}\epsilon^{\mu\nu\sigma\tau} + ig^{\kappa\sigma}\epsilon^{\mu\nu\rho\tau} - ig^{\kappa\tau}\epsilon^{\mu\nu\rho\sigma} + ig^{\mu\nu}\epsilon^{\kappa\rho\sigma\tau} - ig^{\mu\rho}\epsilon^{\kappa\nu\sigma\tau} + ig^{\mu\sigma}\epsilon^{\kappa\nu\rho\tau} - ig^{\mu\tau}\epsilon^{\kappa\nu\rho\sigma} + ig^{\nu\rho}\epsilon^{\kappa\mu\sigma\tau} - ig^{\nu\sigma}\epsilon^{\kappa\mu\rho\tau} + ig^{\nu\tau}\epsilon^{\kappa\mu\rho\sigma} + ig^{\rho\sigma}\epsilon^{\kappa\mu\nu\tau} - ig^{\rho\tau}\epsilon^{\kappa\mu\nu\sigma} + ig^{\sigma\tau}\epsilon^{\kappa\mu\nu\rho}\right)$

Issues with γ^5 in NDR

Assuming that both

$$\{\gamma^5, \gamma^\mu\} = 0,$$
$$\mathrm{Tr}\{\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma\} \neq 0$$

hold in D-dimensions leads to a contradiction. The reason is the assumed cyclicity of the Dirac trace

$$\begin{aligned}
D\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma) &= \mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma\gamma_\tau\gamma^\tau) \\
&= -2g^{\tau\mu}\,\mathrm{Tr}(\gamma^5\gamma^\nu\gamma^\rho\gamma^\sigma\gamma_\tau) + 2g^{\tau\nu}\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\rho\gamma^\sigma\gamma_\tau) \\
&- 2g^{\tau\rho}\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\sigma\gamma_\tau) + 2g^{\tau\sigma}\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma_\tau) \\
&- D\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma) \\
&= -2\,\mathrm{Tr}(\gamma^5\gamma^\nu\gamma^\rho\gamma^\sigma\gamma^\mu) + 2\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\rho\gamma^\sigma\gamma^\nu) \\
&- 2\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\sigma\gamma^\rho) + 2\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma) \\
&- D\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma)
\end{aligned}$$

Issues with γ^5 in NDR

Using that $\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu) = 0$ we have

$$D\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma) = (8-D)\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma)$$

or

$$(4-D)\,\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma) = 0.$$

- This implies that $\mathrm{Tr}(\gamma^5\gamma^\mu\gamma^\nu\gamma^\rho\gamma^\sigma)$ is zero for all $D \neq 4$.
- But if we demand the trace to be meromorphic in D, then the above trace should be zero also for $D = 4$,
- Hence, we cannot recover the 4-dimensional Dirac algebra at $D = 4$.

Non-naive scheme for γ^5

In the Breitenlohner-Maison-'t Hooft-Veltman scheme we are dealing with matrices in D, 4 and D-4 dimensions. Many identities of the BMHV algebra can be proven by by decomposing Dirac matrices into two pieces

$$\dim(\gamma^\mu) = d, \qquad \gamma^\mu = \bar{\gamma}^\mu + \hat{\gamma}^\mu,$$
$$\dim(\bar{\gamma}^\mu) = 4, \qquad g^{\mu\nu} = \bar{g}^{\mu\nu} + \hat{g}^{\mu\nu},$$
$$\dim(\hat{\gamma}^\mu) = d-4. \qquad p^\mu = \bar{p}^\mu + \hat{p}^\mu.$$

(Anti)commuatators between γ's in different dimensions

$$\{\gamma^\mu, \gamma^\nu\} = 2g^{\mu\nu},$$
$$\{\bar{\gamma}^\mu, \bar{\gamma}^\nu\} = \{\gamma^\mu, \bar{\gamma}^\nu\} = 2\bar{g}^{\mu\nu},$$
$$\{\hat{\gamma}^\mu, \hat{\gamma}^\nu\} = \{\gamma^\mu, \hat{\gamma}^\nu\} = 2\hat{g}^{\mu\nu},$$
$$\{\bar{\gamma}^\mu, \hat{\gamma}^\nu\} = 0$$
$$\{\bar{\gamma}^\mu, \gamma^5\} = [\hat{\gamma}^\mu, \gamma^5] = 0,$$
$$\{\gamma^\mu, \gamma^5\} = \{\hat{\gamma}^\mu, \gamma^5\} = 2\hat{\gamma}^\mu\gamma^5 = 2\gamma^5\hat{\gamma}^\mu.$$

Non-naive scheme for γ^5

Contractions of Dirac matrices and vectors with the metric

$$g^{\mu\nu}\gamma_\nu = \gamma^\mu, \qquad g^{\mu\nu}p_\nu = p^\mu,$$
$$\bar{g}^{\mu\nu}\bar{\gamma}_\nu = g^{\mu\nu}\bar{\gamma}_\nu = \bar{g}^{\mu\nu}\gamma_\nu = \bar{\gamma}^\mu, \qquad \bar{g}^{\mu\nu}\bar{p}_\nu = g^{\mu\nu}\bar{p}_\nu = \bar{g}^{\mu\nu}p_\nu = \bar{p}^\mu,$$
$$\hat{g}^{\mu\nu}\hat{\gamma}_\nu = g^{\mu\nu}\hat{\gamma}_\nu = \hat{g}^{\mu\nu}\gamma_\nu = \hat{\gamma}^\mu, \qquad \hat{g}^{\mu\nu}\hat{p}_\nu = g^{\mu\nu}\hat{p}_\nu = \hat{g}^{\mu\nu}p_\nu = \hat{p}^\mu,$$
$$\bar{g}^{\mu\nu}\hat{\gamma}_\nu = \hat{g}^{\mu\nu}\bar{\gamma}_\nu = 0, \qquad \bar{g}^{\mu\nu}\hat{p}_\nu = \hat{g}^{\mu\nu}\bar{p}_\nu = 0.$$

Contractions of the metric with itself

$$g^{\mu\nu}g_{\nu\rho} = g^\mu_\rho \qquad g^{\mu\nu}g_{\mu\nu} = d,$$
$$\bar{g}^{\mu\nu}\bar{g}_{\nu\rho} = g^{\mu\nu}\bar{g}_{\nu\rho} = \bar{g}^{\mu\nu}g_{\nu\rho} = \bar{g}^\mu_\rho \qquad \bar{g}^{\mu\nu}\bar{g}_{\mu\nu} = g^{\mu\nu}\bar{g}_{\mu\nu} = \bar{g}^{\mu\nu}g_{\mu\nu} = 4,$$
$$\hat{g}^{\mu\nu}\hat{g}_{\nu\rho} = g^{\mu\nu}\hat{g}_{\nu\rho} = \hat{g}^{\mu\nu}g_{\nu\rho} = \hat{g}^\mu_\rho \qquad \hat{g}^{\mu\nu}\hat{g}_{\mu\nu} = g^{\mu\nu}\hat{g}_{\mu\nu} = \hat{g}^{\mu\nu}g_{\mu\nu} = d-4,$$
$$\bar{g}^{\mu\nu}\hat{g}_{\nu\rho} = \hat{g}^{\mu\nu}\bar{g}_{\nu\rho} = 0, \qquad \bar{g}^{\mu\nu}\hat{g}_{\mu\nu} = \hat{g}^{\mu\nu}\bar{g}_{\mu\nu} = 0.$$

Contractions of Dirac matrices and vectors with themselves

$$\gamma^\mu\gamma_\mu = D, \qquad p^\mu p_\mu = p^2,$$
$$\bar{\gamma}^\mu\bar{\gamma}_\mu = \gamma^\mu\bar{\gamma}_\mu = \bar{\gamma}^\mu\gamma_\mu = 4, \qquad \bar{p}^\mu\bar{p}_\mu = \bar{p}^\mu p_\mu = p^\mu\bar{p}_\mu = \bar{p}^2,$$
$$\hat{\gamma}^\mu\hat{\gamma}_\mu = \gamma^\mu\hat{\gamma}_\mu = \hat{\gamma}^\mu\gamma_\mu = D-4, \qquad \hat{p}^\mu\hat{p}_\mu = \hat{p}^\mu p_\mu = p^\mu\hat{p}_\mu = \hat{p}^2,$$
$$\bar{\gamma}^\mu\hat{\gamma}_\mu = \hat{\gamma}^\mu\bar{\gamma}_\mu = 0, \qquad \bar{p}^\mu\hat{p}_\mu = \hat{p}^\mu\bar{p}_\mu = 0.$$

Larin-Gorishny-Akyeampong-Delburgo prescription allows one to use anticommuting γ^5 in D-dimensions but compute the chiral traces, such, that the result is expected to be equivalent with the BMHV scheme, if we have only one axial-vector current. The prescription is essentially

- Anticommute γ^5 to the right inside the trace
- Replace $\gamma^\mu \gamma^5$ with $-\frac{i}{6} \varepsilon^{\mu\alpha\beta\sigma} \gamma^\alpha \gamma^\beta \gamma^\sigma$
- Treat $\varepsilon^{\mu\alpha\beta\sigma}$ as if it were D-dimensional, i.e. $\varepsilon^{\mu\alpha\beta\sigma} \varepsilon_{\mu\alpha\beta\sigma} = -D(D^3 - 6D^2 + 11D - 6)$ instead of -24.

Schouten's identity

In an n-dimensional space, a totally antisymmetric tensor with $n+1$ indices vanishes. For example, $e^{ijkl} = 0$ if i,j,k and l are Cartesian indices that run from 1 to 3, because no matter how you choose the values of the indices, you will always have at least two indices with the same value.

- 4D space

$$\varepsilon^{\mu\nu\rho\sigma} p^{\tau} + \varepsilon^{\nu\rho\sigma\tau} p^{\mu} + \varepsilon^{\rho\sigma\tau\mu} p^{\nu} + \varepsilon^{\sigma\tau\mu\nu} p^{\rho} + \varepsilon^{\tau\mu\nu\rho} p^{\sigma} = 0$$

$$\varepsilon^{\mu\nu\rho\sigma} g^{\tau\kappa} + \varepsilon^{\nu\rho\sigma\tau} g^{\mu\kappa} + \varepsilon^{\rho\sigma\tau\mu} g^{\nu\kappa} + \varepsilon^{\sigma\tau\mu\nu} g^{\rho\kappa} + \varepsilon^{\tau\mu\nu\rho} g^{\sigma\kappa} = 0$$

- 3D space

$$\varepsilon^{ijk} p^{l} - \varepsilon^{jkl} p^{i} + \varepsilon^{klj} p^{j} - \varepsilon^{lij} p^{k} = 0$$

$$\varepsilon^{ijk} g^{lm} - \varepsilon^{jkl} p^{im} + \varepsilon^{klj} g^{jm} - \varepsilon^{lij} g^{km} = 0$$

Definitions of the PaVe scalar integrals (LoopTools convention)

$$A_0(m_0) = \mu^{4-D}(4\pi)^{\frac{4-D}{2}} \int \frac{d^D q}{i\pi^{\frac{D}{2}}} \frac{1}{q^2 - m_0^2}$$

$$B_0(p_1, m_0, m_1) = \mu^{4-D}(4\pi)^{\frac{4-D}{2}} \int \frac{d^D q}{i\pi^{\frac{D}{2}}} \frac{1}{(q^2 - m_0^2)((q+p_1)^2 - m_1^2)}$$

$$C_0(p_1, p_2, m_0, m_1, m_2)$$

$$= \mu^{4-D}(4\pi)^{\frac{4-D}{2}} \int \frac{d^D q}{i\pi^{\frac{D}{2}}} \frac{1}{(q^2 - m_0^2)((q+p_1)^2 - m_1^2)((q+p_1+p_2)^2 - m_2^2)}$$

$$D_0(p_1, p_2, p3, m_0, m_1, m_2, m_3)$$

$$= \mu^{4-D}(4\pi)^{\frac{4-D}{2}} \int \frac{d^D q}{i\pi^{\frac{D}{2}}} \frac{1}{(q^2 - m_0^2)((q+p_1)^2 - m_1^2)((q+p_1+p_2)^2 - m_2^2)}$$

$$\times \frac{1}{((q+p_1+p_2+p_3)^2 - m_3^2)}$$

Normalization of the PaVe scalar integrals

Passarino-Veltman scalar functions are normally related to the text book integrals by a factor of $(16\pi^2)/i$, e.g.

$$\mu^{4-D} \int \frac{d^D q}{(2\pi)^D} \frac{1}{q^2 - m_0^2} = \frac{i}{16\pi^2} A_0(m_0)$$

To see this observe that

$$\frac{1}{(2\pi)^D} = \frac{1}{16\pi^2} \frac{1}{2^{D-4}\pi^{D-2}} = \frac{1}{16\pi^2} \frac{4^{\frac{4-D}{2}}}{\pi^{D-2}} = \frac{1}{16\pi^2} \frac{(4\pi)^{\frac{4-D}{2}}}{\pi^{\frac{D}{2}}}$$

However, in FeynCalc the PaVe functions are normalized as $\mu^{4-D} \frac{1}{i\pi^2} \int d^D q (\ldots)$. Hence, we have

$$A_{0,FC}(m_0) = (2\pi)^{D-4} A_0(m_0) = \frac{(2\pi)^D}{i\pi^2} \mu^{4-D} \int \frac{d^D q}{(2\pi)^D} \frac{1}{q^2 - m_0^2}$$

On the other hand, if the prefactor $\frac{1}{(2\pi)^D}$ is implicit (i.e. it is understood but not written down explicitly) in the calculation, then it is enough to perform the replacement

$$A_{0,FC}(m_0) \to \frac{1}{i\pi^2} \mu^{4-D} \int \frac{d^D q}{(2\pi)^D} \frac{1}{q^2 - m_0^2}$$

Belyaev, A., Christensen, N. D., & Pukhov, A. (2012).
CalcHEP 3.4 for collider physics within and beyond the Standard Model.

Breitenlohner, P. & Maison, D. (1977).
Dimensional renormalization and the action principle.
Communications in Mathematical Physics, 52, 11–38.

Chanowitz, M., Furman, M., & Hinchliffe, I. (1979).
The axial current in dimensional regularization.
Nuclear Physics B, 159(1-2), 225–243.

Hahn, T. (2001).
Generating Feynman Diagrams and Amplitudes with FeynArts 3.
Comput.Phys.Commun., 140, 418–431.

Hahn, T. & Perez-Victoria, M. (Comput.Phys.Commun.118:153-165,1999).
Automatized One-Loop Calculations in 4 and D dimensions.
Comput.Phys.Commun., 118:153-165,1999.

Ishikawa, T. et al. (1993).
GRACE manual: Automatic generation of tree amplitudes in Standard Models: Version 1.0.

Jamin, M. & Lautenbacher, M. E. (1993).

TRACER version 1.1.
Computer Physics Communications, 74(2), 265–288.

Jegerlehner, F. (Eur.Phys.J.C18:673-679,2001).
Facts of life with gamma(5).
Eur.Phys.J.C, 18:673-679,2001.

Kreimer, D. (1990).
The γ^5-problem and anomalies — A Clifford algebra approach.
Phys. Lett. B, 237(1), 59–62.

Larin, S. A. (1993).
The renormalization of the axial anomaly in dimensional regularization.
Phys. Lett. B, 303, 113–118.

Mertig, R., Böhm, M., & Denner, A. (1991).
Feyn Calc - Computer-algebraic calculation of Feynman amplitudes.
Computer Physics Communications, 64(3), 345–359.

Passarino, G. & Veltman, M. (1979).
One Loop Corrections for e+ e- Annihilation Into mu+ mu- in the Weinberg Model.
Nucl.Phys., B160, 151.

Patel, H. H. (2015).
Package-X: A Mathematica package for the analytic calculation of one-loop integrals.

Schwinger, J. (1948).
On Quantum-Electrodynamics and the Magnetic Moment of the Electron.
Phys. Rev., 73(4), 416–417.

Smirnov, A. V. (2008).
Algorithm FIRE – Feynman Integral REduction.
JHEP, 0810:107,2008.

t'Hooft, G. & Veltman, M. (1979).
Scalar one-loop integrals.
Nuclear Physics B, 153, 365–401.

Wiebusch, M. (2014).
HEPMath: A Mathematica Package for Semi-Automatic Computations in High Energy Physics.

't Hooft, G. & Veltman, M. (1972).
Regularization and renormalization of gauge fields.
Nuclear Physics B, 44(1), 189–213.